Explizite Differentialgleichungen erster Ordnung

Differentialgleichung, welche durch direkte Integration gelöst werden kann

In[1]:= `DE = y'[x] == 1/(1 + x^2)`

$$\text{DE} = y'[x] == \frac{1}{1+x^2}$$

Out[1]= $y'(x) = \frac{1}{x^2+1}$

- **Wir zeichnen das Richtungsfeld**

In[2]:=
```
DirectionField[DE_, y_[x_], {x_, a_, b_}, {y_, c_, d_}, options___] := Module[{g},
   g = DE[[2]] /. y[x] → y;
   VectorPlot[{1, g}, {x, a, b}, {y, c, d}, options]
  ]
```

In[3]:=
```
plot1 = DirectionField[DE, y[x], {x, -5, 5}, {y, -3, 5}, Frame → True]
```

Out[3]=



- **Wir lösen die Differentialgleichung bzw. das zugehörige Anfangswertproblem**

In[4]:= $\int \frac{1}{1+x^2}\, dx$

Out[4]= $\tan^{-1}(x)$

In[5]:= `DSolve[DE, y [x], x]`

Out[5]= $\{\{y(x) \to c_1 + \tan^{-1}(x)\}\}$

In[6]:= $1 + \int_0^x \frac{1}{1+t^2}\, dt$

Out[6]= $\tan^{-1}(x) + 1$

In[7]:=
```
lösung = DSolve[{DE, y[0] == 1}, y [x], x]
```

Out[7]= $\{\{y(x) \to \tan^{-1}(x) + 1\}\}$

In[8]:=
```
plot2 = Plot[y[x] /. lösung, {x, -5, 5}, PlotStyle → {Thickness[0.01], RGBColor[1, 0, 0]}]
```

Out[8]=



In[9]:=
```
Show[plot2, plot1, PlotRange → {-2, 3}]
```

Out[9]=



In[10]:=
```
sol = NDSolve[{DE, y[0] == 0}, y [x], {x, -1 000 000, 1 000 000}]
```

Out[10]= $\{\{y(x) \to \text{InterpolatingFunction}[(-1.\times 10^6 \quad 1.\times 10^6), <>](x)\}\}$

In[11]:=
```
N[y[x] /. sol[[1]] /. x → 1 000 000]
```

Out[11]= 1.5708

In[12]:= $N\left[\frac{\pi}{2}\right]$

Out[12]= 1.5708

In[13]:= `plot3 = Plot[y[x] /. sol, {x, -5, 5}, PlotStyle → {Thickness[0.01], RGBColor[0, 1, 0]}]`

Out[13]=



In[14]:= `Show[plot3, plot2, plot1, PlotRange → {-3, 5}]`

Out[14]=



Die Differentialgleichung des unbegrenzten Wachstums

In[15]:= `DE = y'[x ] == y[x]`

Out[15]= $y'(x) = y(x)$

- **Wir zeichnen das Richtungsfeld**

In[16]:=
```
plot1 = DirectionField[DE, y[x ], {x, -5, 5}, {y, -1, 5}, Frame → True]
```

Out[16]=



- **Wir lösen die Differentialgleichung bzw. das zugehörige Anfangswertproblem**

In[17]:=
```
DSolve[DE, y [x], x]
```

Out[17]= $\{\{y(x) \to c_1 e^x\}\}$

In[18]:=
```
lösung = DSolve[{DE, y[0] == 1}, y [x], x]
```

Out[18]= $\{\{y(x) \to e^x\}\}$

In[19]:=
```
plot2 = Plot[y[x] /. lösung, {x, -5, 5}, PlotStyle → {Thickness[0.01], RGBColor[1, 0, 0]}]
```

Out[19]=



In[20]:= `Show[plot2, plot1, PlotRange → {-1, 5}]`

Out[20]=



■ allgemeineres Problem

In[21]:= `lösung = DSolve[{y'[x ] == α y[x], y[0] == P}, y [x], x]`

Out[21]= $\{\{y(x) \to P\, e^{\alpha x}\}\}$

Die Differentialgleichung des logistischen Wachstums

In[22]:= `DE = y'[x] == α y[x] - β y[x]^2`

Out[22]= $y'(x) = \alpha\, y(x) - \beta\, y(x)^2$

In[23]:= `plot1 = DirectionField[DE /. {α → 1, β → 1/5}, y[x ], {x, 0, 10}, {y, 0, 5}, Frame → True]`

Out[23]=



In[24]:=
```
plot2 = Plot[Evaluate[y[x] /. DSolve[{DE, y[0] == 1}, y[x], x][[1]] /. {α → 1, β → 1/5}],
  {x, 0, 10}, PlotStyle → {Thickness[0.01], RGBColor[0, 1, 1]}]
```

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

Out[24]=



In[25]:=
```
Show[plot1, plot2]
```

Out[25]=



In[26]:=
```
y[x] /. DSolve[{DE, y[0] == P}, y[x], x][[1]]
```

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

Out[26]=
$$\frac{\alpha P e^{\alpha x}}{\alpha - \beta P + \beta P e^{\alpha x}}$$

■ Schrittweise Lösung

In[27]:=
```
DE
```

Out[27]= $y'(x) = \alpha y(x) - \beta y(x)^2$

In[28]:= `Gleichung =` $\int \frac{1}{\alpha\, y - \beta\, y^2}\, dy == \int 1\, dx$

Out[28]= $\frac{\log(y)}{\alpha} - \frac{\log(\alpha - \beta\, y)}{\alpha} == x$

In[29]:= `Solve[Gleichung, y]`

Out[29]= $\left\{\left\{y \to \frac{\alpha\, e^{\alpha\, x}}{\beta\, e^{\alpha\, x} + 1}\right\}\right\}$

In[30]:= `Apart` $\left[\frac{1}{\alpha\, y - \beta\, y^2}, y\right]$

Out[30]= $\frac{1}{\alpha\, y} - \frac{\beta}{\alpha\, (\beta\, y - \alpha)}$

- **Zugehöriges Anfangswertproblem**

In[31]:= `Gleichung =` $\int_P^y \frac{1}{\alpha\, s - \beta\, s^2}\, ds == \int_0^x 1\, dt$

Out[31]= \$Aborted

In[32]:= `Gleichung = Integrate`$\left[\frac{1}{\alpha\, s - \beta\, s^2}, \{s, P, y\}, \text{GenerateConditions} \to \text{False}\right] == \int_0^x 1\, dt$

Out[32]= $\frac{\log(\alpha - \beta\, P) - \log(P) - \log(\alpha - \beta\, y) + \log(y)}{\alpha} == x$

In[33]:= `Solve[Gleichung, y]`

Out[33]= $\left\{\left\{y \to \frac{\alpha\, P\, e^{\alpha\, x}}{\alpha - \beta\, P + \beta\, P\, e^{\alpha\, x}}\right\}\right\}$

- **Wo ist der Wendepunkt? Wir leiten die Differentialgleichung ab und erhalten**

In[34]:= `D[DE, x]`

Out[34]= $y''(x) == \alpha\, y'(x) - 2\, \beta\, y(x)\, y'(x)$

In[35]:= `zweiteableitung = D[DE, x] /. {Apply[Rule, DE]}`

Out[35]= $y''(x) == \alpha\left(\alpha\, y(x) - \beta\, y(x)^2\right) - 2\, \beta\, y(x)\left(\alpha\, y(x) - \beta\, y(x)^2\right)$

In[36]:= `Map[Factor, zweiteableitung]`

Out[36]= $y''(x) == y(x)\, (\alpha - 2\, \beta\, y(x))\, (\alpha - \beta\, y(x))$

In[37]:= `sol = Solve[zweiteableitung[[2]] == 0, y[x]]`

Out[37]= $\left\{\{y(x) \to 0\}, \left\{y(x) \to \frac{\alpha}{2\, \beta}\right\}, \left\{y(x) \to \frac{\alpha}{\beta}\right\}\right\}$

- **Beispiel 1.4**

In[38]:= `DE = y'[x] ==` $\frac{1}{1 + y[x]^2}$

Out[38]= $y'(x) == \frac{1}{y(x)^2 + 1}$

In[39]:= `DSolve[DE, y[x], x]`

Out[39]= $$\left\{\left\{y(x)\to\frac{\sqrt[3]{\sqrt{(81c_1+81x)^2+2916}+81c_1+81x}}{3\sqrt[3]{2}}-\frac{3\sqrt[3]{2}}{\sqrt[3]{\sqrt{(81c_1+81x)^2+2916}+81c_1+81x}}\right\},\right.$$
$$\left\{y(x)\to\frac{3\left(1+i\sqrt{3}\right)}{2^{2/3}\sqrt[3]{\sqrt{(81c_1+81x)^2+2916}+81c_1+81x}}-\frac{\left(1-i\sqrt{3}\right)\sqrt[3]{\sqrt{(81c_1+81x)^2+2916}+81c_1+81x}}{6\sqrt[3]{2}}\right\},$$
$$\left.\left\{y(x)\to\frac{3\left(1-i\sqrt{3}\right)}{2^{2/3}\sqrt[3]{\sqrt{(81c_1+81x)^2+2916}+81c_1+81x}}-\frac{\left(1+i\sqrt{3}\right)\sqrt[3]{\sqrt{(81c_1+81x)^2+2916}+81c_1+81x}}{6\sqrt[3]{2}}\right\}\right\}$$

Typen expliziter Differentialgleichungen erster Ordnung

- **rechte Seite hängt nur von x ab:**

In[40]:= `DSolve[y'[x] == g[x], y[x], x]`

Out[40]= $$\left\{\left\{y(x)\to\int_1^x g(K[1])\,dK[1]+c_1\right\}\right\}$$

In[41]:= `DSolve[{y'[x] == g[x], y[x0] == y0}, y[x], x]`

Out[41]= $$\left\{\left\{y(x)\to\int_1^x g(K[1])\,dK[1]-\int_1^{x0} g(K[1])\,dK[1]+y0\right\}\right\}$$

- **rechte Seite hängt nur von y ab:**

In[42]:= `DSolve[y'[x] == h[y[x]], y[x], x]`

Out[42]= $$\left\{\left\{y(x)\to\text{InverseFunction}\left[\int_1^{\#1}\frac{1}{h(K[1])}\,dK[1]\,\&\right][c_1+x]\right\}\right\}$$

In[43]:= `DSolve[{y'[x] == h[y[x]], y[x0] == y0}, y[x], x]`

Out[43]= $$\left\{\left\{y(x)\to\text{InverseFunction}\left[\int_1^{\#1}\frac{1}{h(K[1])}\,dK[1]\,\&\right]\left[\int_1^{y0}\frac{1}{h(K[1])}\,dK[1]+x-x0\right]\right\}\right\}$$

- **Separable Differentialgleichung**

In[44]:= `DSolve[y'[x] == g[x] h[y[x]], y[x], x]`

Out[44]= $$\left\{\left\{y(x)\to\text{InverseFunction}\left[\int_1^{\#1}\frac{1}{h(K[1])}\,dK[1]\,\&\right]\left[\int_1^{x} g(K[2])\,dK[2]+c_1\right]\right\}\right\}$$